INGI1131 Practical Exercises
Lab 1: Meeting Oz and its basic data structures

Observations

It is quite possible that you have already worked with Mozart/Oz in a previous course. In that case, this session should not be difficult at all, and therefore, take it as an opportunity to test your knowledge about the language. You can skip the tutorial from exercise 1. Once you solved the other exercises, continue with Lab01extra.pdf.

Interacting with Browser

1. Follow the tutorial at
   http://mozart.github.io/mozart-v1/doc-1.4.0/tutorial/
   Do only chapter 2. Do not continue with chapter 3!

2. Feed a and b and comment about the differences you observe.

   a `{Browse 'Hello Nurse'}`

   b `{Browse "Hello Nurse"}`

   What happen if you configure the browser in order to display strings?
   (Options→Representation→Strings)

3. The browser only allows you to browse a value, but there is no limitation on that. What if you want to browse `X`, `Y` and `Z` at the same time in the following case?

```
local
   X=x Y=y Z=z
in
   {Browse .....}
end
```

   a Try using a list

   b Try using a record

Detecting Warnings and Errors

4. Use the compiler (as mentioned in the tutorial) to detect the warnings in the following piece of code and explain the errors.

   Note: In Oz you can use '%' to comment text. In the following example the text after `F1` and `F2` is commented on purpose.

```
local
   F1 % = fun {$ X} X*X end
   F2 % = fun {$ X} X+X end
in
   {Browse ({F1 3} - {F2 3}) + 4.0}
end
```

Now, remove the comment markers and feed the code to see what happens.

Using Oz as a calculator

5. Suppose you want to calculate the roots of the following function:
   $f(X) = X^2 + 5X - 150$ (Remember the formula ($\frac{-b \pm \sqrt{b^2-4ac}}{2a}$). For this you may either use local variables to store partial values or write the whole thing without using any local variable at all. Try both alternatives.

   Notes:
   The square root of a floating point value is calculated by the function `{Sqrt X}`. Variables can't be reassigned, but they can be masked.

6. Remember that the local statement does two things: it creates a new variable and it sets up an identifier to refer to the variable. The identifier only refers to the variable inside the local statement.

   a Observe the differences between

   ```
   local X in
      X = a
      X = b
   end
   ```

   and

   ```
   local X in
      X = a
      local X in
         X = b
      end
   end
   ```

   Why does the former raise an exception while the latter doesn't?

b What is going to be shown after feeding the code below?

```
local
   Y = 1
   X = 1
in
   local
      Y=2
   in
      {Browse [X Y]}
   end
   {Browse [X Y]}
end
```

Working with Lists and Records

7. Lists are records. The list [a b c] is actually the record
'|'(1:a 2:'|'(1:b 2:'|'(1:c 2:nil)))
Note, however, that, you can write the same expression as
'|'(a '|'(b '|'(c nil)))
Remember also that you can use the infix operator '.' to access a particular field of a record. For instance {`Browse R.a`} will display the value of `R` under feature `a`. Having this in mind, answer the following questions:

   a How many elements does the list `[1 2 3]` have?

   b How many elements does the list `[[1 2 3]]` have?

   c Suppose that `L` is equal to `[[1 2 3]]`. How do you browse the second element of the list inside `L`?

   d How would you solve point 7c if you had N_{th} at your disposition?
   Note: `Nth` is a function on lists. Given a list `Xs` and a position `I`, {`Nth Xs I`} returns the element at position `I` of `Xs`. For instance, if
   {`Browse {Nth [a b c d] 3}`}
   is fed, `c` will be browsed.

   e Browse the following expression:
   '#'(a:5 b:2 3 4) == '#'(1:3 b:2 a:5 2:4)
   What do you get? Why do you get that?

   f Suppose you have that R='#'(a [b '#'(c d) e] f). How do you access `d` through `R`?



(a) Example (b) Exercise

Figure 1: Graphs

8. The Browser offers special support for browsing cyclic records. You can find details in section 2.7 of http://mozart.github.io/mozart-v1/doc-1.4.0/browser/index.html

   a In the following example, `X` is defined as a cyclic record. Feed this piece of code and observe what happens:

   ```
   declare
      X = a(1 X)
      {Browse X}
   ```

   Now select the option 'Minimal Graph' in Options/Representation and feed the code again. How do you explain what you are observing?

   b Let's map graphs to records as follows; each node is a record. The label of the record is the name of the node. The features of the records are the names of the outgoing edges. The value of the record at a particular feature is the destination of the corresponding edge. Under this definition, we can say that the graph in Figure 1a is mapped to the record `R` defined below:

   ```
   declare
      X = e(g h)
      R = a(b(d X) c(X f))
   ```

   Given this mapping, consider the graph in Figure 1b. How would you define the corresponding record in Oz?

9. In each of the following cases, and before feeding the code, say what is going to be shown in the Browser and justify your answer. Run the code to confirm your understanding.

a
```
local X Y in
   X = 1|2|Y
   X = Y
   {Browse X.2.2.1}
end
```

b
```
local X Y Z in
   X = 1|X
   Y = X|Z
   Z = 2|3|4|nil
   {Browse Y.1.2.1}
end
```

c
```
local X Y Z in
   X = a(b X)
   Y = c(X Z)
   Z = d(e f g h)
   {Browse Y.1.2.1}
end
```